
1. Overview

This document lists all changes needed to correct problems in the TTCN implementation of test case 10.5.1b which is part of the LTE/SAE test suite in the ‘iwd-TTCN3-B2013-03_D15wk25’ ATS delivery.

The test case can be demonstrated to run one LTE UE (see section 5). Execution logs are provided as evidence.

3. Verification Test Summary

Test Case:
TC_10_5_1b
Test Group:
ESM_A_Testcases.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk25
System Simulator used:
Anite Conformance Toolset solution
UE’s used:
Intel XMM 7360
Verification Status:
PASS


4. Corrections required for test case 10.5.1b
4.1 Change 1
	Function name
	fl_TC_10_5_1b_Body

	Reason for change
	1. As per "Table 10.5.1b.3.3-2: Message SERVICE REJECT", the value of "T3346 value" needs to be configured instead of "T3342"
2. In Step 5, AT command is not prepared to request connectivity to the same PDN used in step 1 with the NAS signalling low priorityindicator set  to indicate normal priority.
3. According to TS 24.301 Table 8.3.20.1 the ID of Device properties should be “0C” while in the current TTCN implementation it is “0D”.
4. As per prose CR for this TC R5-153663, in Step 7 of the test case procedure a PDN Connection is requested but the procedure to establish the PDN connection is not specified.
Note: The associated prose CR is submitted for discussion in RAN5#68 meeting

	Summary of change
	1. T3342 value is passed as omit from the TC body and "T3346 value" is passed as per spec.
2. AT command updated in step 5.

3. ID of Device properties IE corrected to “0C” in step 7.
4. TTCN changes added for steps 8-9 to complete the PDN Connectivity procedure.

	TTCN module
	ESM_A_Testcases.ttcn

	MCC160
	


Before change

	  function fl_TC_10_5_1b_Body() runs on EUTRA_PTC

  {

    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;

    var NAS_MSG_Indication_Type v_NasInd;

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var octetstring v_APN_Value := tsc_APN1; // APN-1(New PDN name), different from default

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NasCount_Type v_NasCountUL;

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_CellA);

    var B1_Type v_DeviceProperties_1 := '1'B ;

    var B1_Type v_DeviceProperties_0 := '0'B ;

    timer t_T3346:= f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 60.0);

    // Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 1" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String);

    //@siclog "Step 1A- 2" siclog@

    // RRC establishment cause shall be set to Delay tolerant according to 24.301 Table D.1.1

    v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                              tsc_RRC_TI_Def,

                                              cr_EstablishmentCause_delayTolerantAccess, //  @sic R5-142889 sic@

                                              cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                cr_EXT_SERVICE_REQUEST_DeviceProperties(v_SecurityParams.KSIasme,

                                                                                                        bit2oct(v_GutiParams.M_TMSI),

                                                                                                        cr_DeviceProperties('D'H, v_DeviceProperties_1))));

    //+  The SS transmits a SERVICE REJECT message with the EMM cause set to #22

    //+   (UE identity cannot be derived by the network).

    //@siclog "Step 3" siclog@

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                cs_508_SERVICE_REJECT(tsc_Cause_Congestion,

                                                                      cs_GprsTimer_tv('5F'O, '001'B, '00101'B))))); //@sic R5-142874 sic@ @sic R5-152082 sic@

    //@siclog "Step 4" siclog@

    t_T3346.start;

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA); //Release cause= other

    //@siclog "Step 5" siclog@

    //Cause the UE to request connectivity to the same PDN as used in step 1, with the NAS signalling low priority indicator set to indicate normal priority.

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String);

    //@siclog "Step 5A- 6" siclog@

    v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                              tsc_RRC_TI_Def,

                                              cr_EstablishmentCause_MoData, //  @sic R5-142889 sic@

                                              cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                cr_EXT_SERVICE_REQUEST_DeviceProperties(v_SecurityParams.KSIasme,

                                                                                                        bit2oct(v_GutiParams.M_TMSI),

                                                                                                        cr_DeviceProperties('D'H, v_DeviceProperties_0))));

    //@siclog "Step 6A" siclog@

    // The SS establishes a DRB associated with the default EPS bearer context activated

    //  @sic R5-142889 sic@

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(v_RRC_TI,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                             f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA))),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 7" siclog@

    alt{

    [] SRB.receive(car_SRB2_NasPdu_IND (eutra_CellA,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN_DeviceProperties(f_GetPdnType(),

                                                                                                        cr_AccessPointName(v_APN_Value),

                                                                                                        cr_DeviceProperties('D'H, v_DeviceProperties_0)))))

      {

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");

        t_T3346.stop;

      }

    [] t_T3346.timeout

       {

         f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "T3396 timeout, No PDN as specified in Table 10.5.1b.3.3-5 received");

       }

    }

  }  


After change

	     function fl_TC_10_5_1b_Body() runs on EUTRA_PTC

  {

    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;

    var NAS_MSG_Indication_Type v_NasInd;

    var ProcedureTransactionIdentifier v_PTId_UE;     //Change

    var template (value) AccessPointName v_AccessPointName;  //Change

    var template (omit) ProtocolConfigOptions v_Pco;   //Change

    var PDN_Address v_PDN_Address2;    //Change

    var SRB_COMMON_IND v_ReceivedAsp;  //Change
    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var octetstring v_APN_Value := tsc_APN1; // APN-1(New PDN name), different from default

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NasCount_Type v_NasCountUL;

    var PDN_CONNECTIVITY_REQUEST v_PDNConnReq;  //Change
    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var template (value) EUTRA_FDD_TDD_CellInfo_Type v_EUTRA_FDD_TDD_Info := f_EUTRA_CellInfo_GetFDD_TDD_Info (eutra_CellA);

    var B1_Type v_DeviceProperties_1 := '1'B ;

    var B1_Type v_DeviceProperties_0 := '0'B ;

    timer t_T3346:= f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 60.0);

    // Cause the UE to request connectivity to an additional PDN.

    //@siclog "Step 1" siclog@

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String);

    //@siclog "Step 1A- 2" siclog@

    // RRC establishment cause shall be set to Delay tolerant according to 24.301 Table D.1.1

    v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                              tsc_RRC_TI_Def,

                                              cr_EstablishmentCause_delayTolerantAccess, //  @sic R5-142889 sic@

                                              cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                cr_EXT_SERVICE_REQUEST_DeviceProperties(v_SecurityParams.KSIasme,

                                                                                                        bit2oct(v_GutiParams.M_TMSI),

                                                                                                        cr_DeviceProperties('D'H, v_DeviceProperties_1))));

    //+  The SS transmits a SERVICE REJECT message with the EMM cause set to #22

    //+   (UE identity cannot be derived by the network).

    //@siclog "Step 3" siclog@

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(tsc_SHT_NoSecurityProtection,

                                                cs_508_SERVICE_REJECT(tsc_Cause_Congestion, -, // GPRs timer value should be omit 

                                                                      cr_GPRS_Timer2_3('5F'O, '000'B, '11110'B))))); //@sic R5-142874 sic@ @sic R5-152082 sic@     

    //@siclog "Step 4" siclog@

    t_T3346.start;

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA); //Release cause= other

    //@siclog "Step 5" siclog@

    //Cause the UE to request connectivity to the same PDN as used in step 1, with the NAS signalling low priority indicator set to indicate normal priority.

    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String, -, -, 1);
    //@siclog "Step 5A- 6" siclog@

    v_NasInd:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                              tsc_RRC_TI_Def,

                                              cr_EstablishmentCause_MoData, //  @sic R5-142889 sic@

                                              cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                                cr_EXT_SERVICE_REQUEST_DeviceProperties(v_SecurityParams.KSIasme,

                                                                                                        bit2oct(v_GutiParams.M_TMSI),

                                                                                                        cr_DeviceProperties('D'H, v_DeviceProperties_0)))); //Change from 'D'H to 'C'H IE in Device property because C IE is valid for ESM message like PDN Req(Ref 3GPP 24.301 section 8.3.20

    //@siclog "Step 6A" siclog@

    // The SS establishes a DRB associated with the default EPS bearer context activated

    //  @sic R5-142889 sic@

    v_NasCountUL := v_NasInd.SecurityProtection.NasCount;

    v_SecurityParams := f_EUTRA_Authentication_InitAS(v_SecurityParams, v_NasCountUL);

    v_SecurityParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_SecurityParams, v_NasCountUL);

    f_EUTRA_Security_Set(v_SecurityParams);

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_Common(v_RRC_TI,

                                                                        omit,

                                                                        omit,

                                                                        cs_508_RadioResourceConfigDedicated_SRB2( {cs_508_DRB_ToAddMod_DEFAULT_AM (tsc_DRB1)},

                                                                                                                  cs_508_MAC_MainConfig_Explicit_RBC_DrxL,

                                                                                                                  cs_508_PhysicalConfigDedicated_Default_RBC(v_EUTRA_FDD_TDD_Info.cqi_ReportConfig,

                                                                                                                                                             v_EUTRA_FDD_TDD_Info.soundingRS_UL_Config,

                                                                                                                                                             f_EUTRA_CellInfo_GetAntennaInfoDedicated (eutra_CellA))),

                                                                        omit)));

    // receive RRCConnectionReconfigurationComplete

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_CellA, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

    //@siclog "Step 7" siclog@

    alt{

    [] SRB.receive(car_SRB2_NasPdu_IND (eutra_CellA,

                                        cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_PDNConnReqAdditionalPDN_DeviceProperties(f_GetPdnType(),

                                                                                                        cr_AccessPointName(v_APN_Value),

                                                                                                        cr_DeviceProperties('C'H, v_DeviceProperties_0))))) -> value v_ReceivedAsp
      {

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 7");

        t_T3346.stop;

      }

    [] t_T3346.timeout

       {

         f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "T3396 timeout, No PDN as specified in Table 10.5.1b.3.3-5 received");

       }

    }

    //Change for completing  PDN connection procedure at Step 7. // Step 8 and 9 added as RAN5 prose CR

      v_PDNConnReq := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.pDN_CONNECTIVITY_REQUEST;

      v_PTId_UE := v_PDNConnReq.procedureTransactionIdentifier;

      v_AccessPointName := cs_AccessPointName(v_APN_Value);

      v_Pco := v_PDNConnReq.protocolConfigurationOptions;

    // Change from DRB1 to DRB2 to handle U Plane data

    f_EUTRA_StartIPfor2ndPDN(eutra_CellA); 

    //@siclog "Step 8" siclog@

    // The SS transmits an ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message with IE EPS Bearer Identity set to new EPS bearer context.

    // Ask UE to get its additional IP adress via DHCP or ICMP

    v_PDN_Address2 := f_EUTRA_GetPDNAddress (f_CheckPCOforIPallocationViaNas (v_Pco), PDN_2);

    f_EUTRA_TxActivateDefaultEpsBearerCtxtReq_DrbReconfig (eutra_CellA,

                                                           v_RRC_TI,

                                                           v_EpsBearerId2, // Default bearer ID to additional PDN

                                                           v_PTId_UE,  // use value as defined by the UE

                                                           v_AccessPointName,

                                                           v_PDN_Address2,

                                                           f_GetDefaultProtocolConfigOptions(v_Pco)); 

    // Check: Does the UE transmit an ACTIVATE DEFAULT EPS BEARER CONTEXT ACCEPT message for the additional default EPS bearer?

    //@siclog "Step 9" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ActivateDefEpsBearerContextAccept(v_EpsBearerId2))));  

    // 1.2s delay acc. to 36.508 cl. 4.5A.1 to allow user-plane signalling

    f_EUTRA_DelayForUserPlaneSignalling(); 

  }


5. Execution Log Files
1) Intel XMM 7360
The Intel XMM 7360 UE passed this Rel-11 test case on Anite Conformance Toolset Solution in LTE FDD path on LTE band 4. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file: 
TC_10_5_1b_Log.html
References

	[1]
	R5s150779: Supporting information for agreement of TC 10.5.1b in LTE-FDD mode. This archive comprises: html and text format execution log files as well as PICS/PIXIT settings.

	
	


